# Maintainer: Sven-Hendrik Haase <sh@lutzhaase.com>
# Contributor: Thomas S Hatch <thatch45@gmail.com>
# Contributor: Geoffroy Carrier <geoffroy@archlinux.org>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>

pkgbase=protobuf
pkgname=('protobuf' 'python2-protobuf')
pkgver=2.5.0
pkgrel=3
pkgdesc="A way of encoding structured data in an efficient yet extensible format"
arch=('i686' 'x86_64')
url="http://code.google.com/p/protobuf/"
license=('APACHE')
depends=('gcc-libs' 'zlib')
makedepends=('python2' 'setuptools')
options=(!libtool)
source=(http://$pkgname.googlecode.com/files/$pkgname-$pkgver.tar.bz2)
md5sums=('a72001a9067a4c2c4e0e836d0f92ece4')

build() {
	cd $srcdir/$pkgname-$pkgver
	./autogen.sh
	./configure --prefix=/usr --disable-static
}

package_protobuf() {
	cd $srcdir/$pkgname-$pkgver
	make DESTDIR=$pkgdir install
}

package_python2-protobuf() {
	pkgdesc="protobuf python API"
	depends=("python2" "protobuf=${pkgver}")
    provides=("protobuf-python")
    conflicts=("protobuf-python")
    replaces=("protobuf-python")

	cd $srcdir/protobuf-$pkgver/python
	python2 setup.py install --prefix=/usr --root $pkgdir
}
